CS 360
Programming Languages
First-class functions


THE DAWN OF ~~MAN~~
Functions

An Example

- What if we wanted to add up all the numbers from a to b?

$$\sum_{i=a}^{b} i$$

```
(define (sum a b)
  (if (> a b)
    0
    (+ a
       (sum (+ a 1) b))))
```

An Example

- What if we wanted to add up all the **squares** numbers from a to b?

$$\sum_{i=a}^{b} i^2$$

```
(define (sum a b)
  (if (> a b)
    0
    (+ (expt a 2)
       (sum (+ a 1) b))))
```

An Example

- What if we wanted to add up all the **absolute values of the** numbers from a to b?

$$\sum_{i=a}^{b} |i|$$

```
(define (sum a b)
  (if (> a b)
    0
    (+ (abs a)
       (sum (+ a 1) b))))
```

These functions are all very similar

- All three of these functions differ only in how the sequence of integers from a to b are transformed before they are all added together.
- The adding process itself is identical in all of the functions:

```
(define (sum-something a b)
  (if (> a b)
    0
    (+ (do something to a)
       (sum-something (+ a 1) b))))
```

- What if there were a general sum function that could sum up any sequence of this form?

A function that takes a function

- Imagine a function that could take another function as an argument:

```
(define (sum-any func a b)
  (if (> a b)
    0
    (+ (func a)
       (sum-any func (+ a 1) b))))
```

Sum-any in action!

```
(sum-any sqrt 1 10)
  => sqrt(1) + sqrt(2) + sqrt(3) + …
  => about 22.5

(define (square x) (* x x))
(sum-any square 1 4)
  => 1^2 + 2^2 + 3^2 + 4^2 => 1 + 4 + 9 +
16 => 30

(define (identity x) x)
(sum-any identity 1 4)
  => 10
```

How to use sum-any

- You can put the name of any function in place of **`sqrt`**, **`square`**, or **`identity`**, and **`sum-any`** will compute

  f(a) + f(a + 1) + f(a + 2) + … + f(b)

  – Provided **`f`** is a function of a single numeric argument.

- What if you want to compute f(a^2/2) + f((a+1)^2/2) + …
  – Fine to do:

```
(define (silly-function x) (/ (* x x) 2))
(sum-any silly-function 1 10)
```

- Wouldn't it be nicer if we didn't have to name that silly function?

Anonymous Functions

- Functional programming languages allow us to create functions without names.
- In Racket, we use the keyword **`lambda`** for this:

  `(lambda (arg1 arg2…) body)`

- This expression represents an *anonymous function*.
  - Kind of like a "function literals."

Aside: lambda calculus

- Formal system for computation based on function abstraction and application.
- *Church-Turing thesis* (1936-37) proved lambda calculus is equivalent in power to Turing machines.

Alonzo Church

Anonymous Functions

- Use an anonymous function when you need a "temporary" function:

```
(sum-any (lambda (x) (/ (* x x) 2)) 1 10)
```

is better style than

```
(define (silly-function x) (/ (* x x) 2))
(sum-any silly-function 1 10)
```

- Compare:

```
(sum-any (lambda (x) (* x x)) 1 10)
```

and

```
(define (square (x) (* x x))
(sum-any square 1 10)
```

Using anonymous functions

- Most common use: Argument to a higher-order function
  - Don't need a name just to pass a function

```
(define (triple x) (* 3 x); named version

(lambda (x) (* 3 x))       ; anonymous version
```

- But: Cannot use an anonymous function for a recursive function
  - Because there is no name for making recursive calls

Named functions vs anonymous functions

- Named functions are mostly indistinguishable from anonymous functions.
- In fact, naming a function with **`define`** uses the anonymous form behind the scenes:

```
(define (func arg1 arg2 …) expression)
```

is converted to:

```
(define func (lambda (arg1 arg2 …) expression))
```

- It is poor style to define unnecessary functions in the global (top-level) environment
  - Use either nested defines, or anonymous functions.

Higher-order functions

- A ***higher-order function*** is a function that either takes a function (or more than one function) as an argument, or returns a function as a return value.
- Possible because functions are ***first-class values*** (or ***first-class citizens***), meaning we can use a function wherever we use a value.
  - First class citizens can be arguments to functions, returned from functions, bound to variables, and stored in data structures.
  - In Racket, a function can be stored anywhere any other data type would be stored.
- Most common use is as an argument / result of another function




Higher-order functions

- Let's see another:

```
(define (do-n-times func n x)
  (if (= n 0) x
    (do-n-times func (- n 1) (func x))))
```

- This function computes f(f(f…(x))), where the number of applications of f is n.

Some uses for do-n-times

- Get-nth:
  - ```
    (define (get-nth lst n)
      (car (do-n-times cdr n lst)))
    ```
- Exponentiation:
  - ```
    (define (power x y) ; raise x to the y power
      (do-n-times (lambda (a) (* x a)) y 1))
    ```
- Note how in the exponentiation example, the anonymous function uses variable x from the outer environment.
  - Couldn't do that without being able to nest functions.
- Note how do-n-times can work with any data type (e.g., lists, numbers…)

A style point

Compare:

```
(if x #t #f)
```

With:

```
(lambda (x) (f x)
```

So don’t do this:

```
(do-n-times (lambda (x) (cdr x)) 3 '(2 4 6 8))
```

When you can do this:

```
(do-n-times cdr 3 '(2 4 6 8))
```

What does this function do?

```
(define (mystery lst)
  (if (null? lst) '()
    (cons (car lst) (mystery (cdr lst)))))
```

Map

```
(define (map func lst)
  (if (null? lst) '()
    (cons (func (car lst)) (map func (cdr lst)))))
```

Map is, without doubt, in the higher-order function hall-of-fame

- The name is standard (same in most languages)
- You use it *all the time* once you know it: saves a little space, but more importantly, *communicates what you are doing*
- Built into Racket, so you don't have to include this definition in programs that use map.

Filter

```
(define (filter func lst)
  (cond ((null? lst) '())
        ((func (car lst))
            (cons (car lst) (filter func (cdr lst))))
        (#t
            (filter func (cdr lst)))))
```

- Filter is also in the hall-of-fame
  - So use it whenever your computation is a filter